

Supplementary Material to the paper "Multiparameter actuation of a neutrally - stable shell : a flexible gearless motor" by W.Hamouche, C.Maurini, S.Vidoli, A.Vincenti Last Modified 22 mai 2017

Laminates theory and uniform curvature model: precession experiment

```
In[1]:= Clear["Global`*"];  
ToVoigt[{{k11_, k12_}, {k12_, k22_}}] := {k11, k22, 2 k12};  
ed[phi_] = {Cos[phi], Sin[phi]};
```

Properties of the copper disk

Geometry and elastic properties (MKS units)

```
In[4]:= radius = 0.1225; S = Pi radius^2; h = 0.0003;  
nu = 0.33;  
Y = 124 * 10^9; G = 44.86 * 10^9;
```

Plane-stress stiffness matrix (we align global reference with the material reference of the copper disk, which is not isotropic because of the shear stiffness)

```
In[6]:= Qdisk = Y {{1, nu, 0}, {nu, 1, 0}, {0, 0, G / Y}} / (1 - nu^2)  
Out[6]:= {{1.39154 * 10^11, 4.59208 * 10^10, 0.},  
{4.59208 * 10^10, 1.39154 * 10^11, 0.}, {0., 0., 5.03423 * 10^10}}
```

Properties of the MFC actuators

Geometry and elastic properties

```
In[7]:= hpzt = 0.0003; Spzt = 2 * 0.085 * 0.014;
Y1 = 30.34 * 10^9; Y2 = 15.86 * 10^9;
nu12 = 0.31;
G12 = 5.515 * 10^9;
```

Equivalent thermal expansion coefficient to model the piezoelectric effect

```
In[9]:= a1 = 0.72 * 10^(-6); a2 = -0.38 * 10^(-6);
```

Plane-stress stiffness matrix for a piezoelectric layer rotated by an angle q

```
In[10]:= cos = Cos[q]; sin = Sin[q];
Q11 = Y1 / (1 - nu12^2 Y2 / Y1);
Q22 = Y2 Q11 / Y1;
Q12 = nu12 Q22;
Q66 = G12;
Qb11 = Q11 cos^4 + 2 (Q12 + 2 Q66) sin^2 cos^2 + Q22 sin^4;
Qb12 = (Q11 + Q22 - 4 Q66) sin^2 cos^2 + Q12 (sin^4 + cos^4);
Qb22 = Q11 sin^4 + 2 (Q12 + 2 Q66) sin^2 cos^2 + Q22 cos^4;
Qb16 = (Q11 - Q12 - 2 Q66) sin cos^3 + (Q12 - Q22 + 2 Q66) sin^3 cos;
Qb26 = (Q11 - Q12 - 2 Q66) sin^3 cos + (Q12 - Q22 + 2 Q66) sin cos^3;
Qb66 = (Q11 + Q22 - 2 Q12 - 2 Q66) sin^2 cos^2 + Q66 (sin^4 + cos^4);
Qbpt = {{Qb11, Qb12, Qb16}, {Qb12, Qb22, Qb26}, {Qb16, Qb26, Qb66}};
```

Equivalent thermal expansion coefficients for a piezoelectric layer rotated by an angle q

```
In[19]:= axx = a1 cos^2 + a2 sin^2;
ayy = a1 sin^2 + a2 cos^2;
axy = (a1 - a2) sin cos;
```

Classical Laminate Theory (CLT)

For the copper disk without the actuator we get

```
In[22]:= z[2] = h / 2; z[1] = h / 2 + hpzt; z[3] = -h / 2;
Ddisk = (1 / 3) Qdisk (z[2]^3 - z[3]^3)
```

```
Out[23]= {{0.313096, 0.103322, 0.}, {0.103322, 0.313096, 0.}, {0., 0., 0.11327}}
```

For the regions including the base copper disk and the piezoelectric MFC actuators we have a two-layer layout

We first calculate the extension-to-bending coupling matrix B and choose the mid-plane coordinate z_c , so as to minimize the norm of the coupling, that we will not consider later.

```

In[24]:= z[2] = zc + h / 2; z[1] = zc + h / 2 + hpzt; z[3] = zc - h / 2;
Biso = (1 / 2) Qdisk (z[3]^2 - z[2]^2);
q1 = Pi / 2; q2 = Pi / 2 + 2 Pi / 3; q3 = Pi / 2 - 2 Pi / 3;
Bpzt[q_] = (1 / 2) Qbpzt (z[2]^2 - z[1]^2);
Btot = Expand[Biso + Bpzt[0]];
zcs = zc /. Last[Minimize[Flatten[Btot].Flatten[Btot] // Simplify, z]];

```

```
Out[29]= -0.0000431553
```

Using this new midplane we calculate the bending stiffness in the region composed of two layers

```

In[30]:= z[2] = zcs + h / 2; z[1] = zcs + h / 2 + hpzt; z[3] = zcs - h / 2;
Dpzt[q_] = (1 / 3) Qbpzt (z[1]^3 - z[2]^3);

```

And the bending moment generated by the applied voltage though piezoelectric effect

```

In[32]:= Mpzt[q_, V_] = -V Qbpzt.{axx, ayy, 2 axy} Integrate[z, {z, z[1], z[2]}];

```

Stiffness of the uniform curvature model

We calculate the bending stiffness of the uniform curvature model of the shell by averaging out the stiffness of the different regions

```

In[33]:= q1 = Pi / 2; q2 = -Pi / 6; q3 = Pi / 6;
Dmpzt = (Spzt / S) (Dpzt[q1] + Dpzt[q2] + Dpzt[q3]);
Duc = (Ddisk + Dmpzt) S

```

```
Out[35]= {{0.0182687, 0.00600488, 0.}, {0.00600488, 0.0182687, 0.}, {0., 0., 0.00652713}}
```

While, for an homogeneous disk

```

In[36]:= Ducdisk = Ddisk S

```

```
Out[36]= {{0.0147605, 0.00487095, 0.}, {0.00487095, 0.0147605, 0.}, {0., 0., 0.00533995}}
```

We can calculate the equivalent nondimensional shear stiffness for the two matrices

```

In[37]:= CalculateAlpha[D_] := D[[3, 3]] / (D[[1, 1]] (1 - D[[1, 2]] / D[[1, 1]]) / 2)
CalculateAlpha[Ducdisk]
CalculateAlpha[Duc]

```

```
Out[38]= 1.07992
```

```
Out[39]= 1.06445
```

Inelastic curvatures in the uniform curvature model induced by the piezoelectric actuators

```

In[40]:= Mmpzt = Expand[(Spzt) (Mpzt[q1, V1] + Mpzt[q2, V2] + Mpzt[q3, V3])];
kpzt = -Chop[Expand[Inverse[Duc].Mmpzt]]
{χ2, χ1} = Most[kpzt /. {V1 → 1, V2 → 0, V3 → 0}]

```

```

Out[41]= {0.000107274 V1 - 0.000157979 V2 - 0.000157979 V3,
-0.000246397 V1 + 0.0000188563 V2 + 0.0000188563 V3, 0.000287744 V2 - 0.000287744 V3}

```

```
Out[42]= {0.000107274, -0.000246397}
```

The imposed voltage and the corresponding inelastic curvature


We assume the following inelastic curvature and piezoelectric voltages

```
In[43]:= InelasticCurvatureWithPiezoUC = -3.29;
kbv = {1., 1., 0} * InelasticCurvatureWithPiezoUC + kpzt
```

```
Out[44]:= {-3.29 + 0.000107274 V1 - 0.000157979 V2 - 0.000157979 V3,
-3.29 - 0.000246397 V1 + 0.0000188563 V2 + 0.0000188563 V3,
0. + 0.000287744 V2 - 0.000287744 V3}
```

Time laws for the applied voltages

```
In[45]:= Vrulepos =
{V1 → V (1 + Cos[t]) / 2, V2 → V (1 + Cos[t + 4 q2]) / 2, V3 → V (1 + Cos[t + 4 q3]) / 2};
Plot[Evaluate[{V1, V2, V3} /. Vrulepos /. {V → 1}, {t, 0, 2 Pi}]]
```



Numerical solution of the UC model

```
In[47]:= Off[FindMinimum::lstol]; Off[FindMinimum::fmgz];
kflatv = ToVoigt[Outer[Times, ed[phi], ed[phi]] k];
Utot[V_, t_, k_, phi_] =
Chop[Expand[1/2 (Duc. (kflatv - kbv)). (kflatv - kbv) /. Vrulepos]];
Solver[V_, t_, ktt_, phitt_] :=
FindMinimum[Utot[V, t, k, phi], {k, ktt}, {phi, phitt}];
dt = Pi / 100.; k0 = k /. Last[Minimize[Utot[0, 0, k, 0.1], k]];
phibphiklist[V_] := Module[{kt = k0, phit = 0.},
Table[sol = Solver[V, t, kt, phit];
{phit, kt} = Chop[{phi, k} /. Last[sol]], {t, 0, 2 Pi, dt}]]
```

Assign the voltage $V \in [0, 2000]$ and see the time histories of k and φ

```

In[53]:= V = 1500.;
{phiUC, kUC} = Transpose[phibphiklist[V]];
GraphicsRow[
  {ListPlot[kUC, Frame → True, FrameLabel → {"step", "k"}, PlotRange → {-4, -5}],
   ListPlot[Mod[phiUC, Pi], Frame → True,
    FrameLabel → {"step", " $\phi$ "}, PlotRange → {0, Pi}]], ImageSize → 600]

```

